

CS 182 Spring 2004. Assignment 9 Solution.

Construction Grammar Learning

The solutions here are model answers only. There are a number of equally good answers to these questions.

Part I. Analysis: Using a9.ecg, analyze the following sentences:

(a) *I will bring you the cup*

What is the difference between this analysis and the one for *I will bring the cup*? How is this difference captured in the grammar?

Main difference: The **transitive** construction is used in *I will bring the cup* and the **ditransitive** construction is used in *I will bring you the cup* (as chart entry 0 shows). There is also a **transfer** schema in the SemsSpec, where the speaker is the sender, the addressee is the receiver, and the cup is the object being transferred.

This difference is captured by the **transfer** schema evoked by the ditransitive construction, as well as the appropriate identification constraints.

(b) *I will bring the cup here*

Does the analyzer find a spanning analysis? Why or why not? How does the analysis account for the destination of the bring action, if at all?

The analyzer does not find a spanning analysis – it matches *I will bring the cup* to the transitive construction, but *here* is not covered. This is evident by the fact that (1) the chart entry 0 shows a span of [0,2] instead of [0,3], and (2) the direction in the **cause-move** schema is not bound to the **here** schema.

To capture this correctly, a new constructional constituent, *path*, needs to be added to the transitive construction. The meaning pole of *path* (*path.m*) then needs to be bound to *v.m.direction*.

(c) * *Nomi give I the ball*

Does the analyzer find a spanning analysis here? If it does not, explain why. If it does, propose a change to the grammar that will exclude this sentence from being admitted as grammatical.

The analyzer does find a spanning analysis, even though this sentence is ungrammatical. This is due to the fact that the ditransitive construction as written does not place any restriction on the case marking of the receiver. To correct this (and agreement in general), the form poles of pronoun constructions need to be augmented with agreement features, such as case. Constructions that use pronouns then need to place identification constraints on case.

For the **I** construction, the case marking is assigned to be nominative. Then for the ditransitive construction, the case marking of the receiver needs to be restricted to accusative. The minimal changes are highlighted below:

```
schema Schematic-Form
  subcase of formpoleschema
  roles
    orth
    case
```

```
construction I-Cn
  level 0
  subcase of Ref-Expr
  form : Word
  self.f.orth <-- "I"
```

```

    self.f.case <-- nominative
meaning
  evokes Speaker as n
  self.m.category <--> n
  self.m.resolved-ref <--> n

construction DiTransitive-Cn
  level 2
  constructional
  constituents
    agt : Ref-Expr
    v : Cause-Motion-Verb
    rec : Ref-Expr
    pat : Ref-Expr
  form
    agt.f before v.f
    v.f before rec.f
    rec.f before pat.f
    rec.f.case <-- accusative
meaning
  evokes Transfer as trans
  v.m.agent <--> agt.m.resolved-ref
  v.m.patient <--> pat.m.resolved-ref
  trans.sender <--> agt.m.resolved-ref
  trans.receiver <--> rec.m.resolved-ref
  trans.object <--> pat.m.resolved-ref

```

Part II. Learning

(a) For each utterance,

- i. analyze the sentence using `nomi.ecg`, and compare the `SemSpec` against what is available in the context,
- ii. state any relational mapping operations that may be performed and write down the new construction in ECG notation (if any)

Depending on learning parameters, the most basic case is to map one meaning relation to one form relation, i.e. chunking two constituents at a time. However, there is no inherent reason computationally or psychologically why this has to be the case. It is quite reasonable to children to chunk multiple words at a time (in a rote-learning fashion). Cognitive capacity places an upper bound on the amount that can be learned from one instance, but we do not know what it is empirically, and this undoubtedly varies by individual. In the system this is modeled as a learning parameter. For this assignment, chunking 2 to 3 words at a time is fine.

1. (Father) I will bring you some juice

Possibilities include: `Bring-You-Cn`, `Bring-You-Juice-Cn`, `Bring-Juice-Cn`, `I-Bring-Cn`, `I-Bring-You-Cn`

Example:

```

construction Bring-You-Cn
  level 2
  constructional
  constituents
    bring : Bring-Cn
    you : You-Cn
  form
    bring.f before you.f
  meaning
    bring.m.direction <--> you.m

```

2. (Father) I will get a cup for you Nomi

Possibilities include: I-Get-Cn, I-Get-Cup-Cn, Get-Cup-Cn, Get-Cup-You-Cn, Get-You-Cn

Example:

```
construction I-Get-Cup-Cn
  level 2
  constructional
  constituents
    i : I-Cn
    get : Get-Cn
    cup : Cup-Cn
  form
    i.f before get.f
    get.f before cup.f
  meaning
    get.m.causer <--> i.m
    get.m.mover <--> cup.m
```

3. (Mother) Nomi you can get a book and bring it back here

Skipping of two constituents has to be allowed in order for this sentence to be analyzed successfully.

Possibilities include: You-Get-Cn, You-Get-Book-Cn, Get-Book-Cn, Bring-Here-Cn

```
construction You-Get-Cn
  level 2
  constructional
  constituents
    you : You-Cn
    get : Get-Cn
  form
    you.f before get.f
  meaning
    get.m.causer <--> you.m
```

Depending on the parameters of the cost function, the child may erroneously learn a Book-Bring-Here-Cn, since the child only understood *Nomi you get book bring here* and the context supplies a meaning of bringing the book here. This, in effect, is a piece of noisy data that the learner unfortunately picks up on. However, the learner will eventually see enough examples of good data to average out the noise – this erroneous construction will rarely get reinforced by use.

4. (Mother) now get the red ball Nomi

Possibilities include: Get-Red-Ball-Cn, Get-Red-Ball-Nomi-Cn. Notice that there is already a Red-Ball-Cn.

```
construction Get-Red-Ball-Nomi-Cn
  level 2
  constructional
  constituents
    get : Get-Cn
    rb : Red-Ball-Cn
    nomi : Nomi-Cn
  form
    get.f before rb.f
    rb.f before nomi.f
  meaning
    get.m.causer <--> nomi.m
    get.m.mover <--> rb.m
```

If prosodic information is also supplied, the learner may eventually learn an imperative construction.

5. (Father) would you get that and give it to me please

The learner has not been able to understand the utterance – this is not unusual.

6. (Mother) get the baby and bring her to bed

Possibilities include: Get-Baby-Cn, Bring-Bed-Cn

```
construction Get-Baby-Cn
  level 2
  constructional
  constituents
    get : Get-Cn
    baby : Baby-Cn
  form
    get.f before baby.f
  meaning
    get.m.mover <--> cup.m
```

(b) At the end of the six utterances, state two learning by merging operations that can be performed on the new sets of constructions Nomi just acquired.

Example 1: Merging Get-Red-Ball-Nomi-Cn, Get-Baby-Cn

The learner can generalize either to the more specific Get-Toy-Cn or the more general Get-Object-Cn. On the form side both constructions has an ordering constraint between *get* and the object. Additionally, both Red-Ball-Cn and Baby-Cn are subcase of Ref-Expr. On the meaning side, Baby and Ball are both subcase of Toy, which is then subcase of Manipulable-Object.

```
construction Get-Toy-Cn
  level 2
  constructional
  constituents
    get : Get-Cn
    toy : Ref-Expr
  form
    get.f before toy.f
  meaning
    evokes Toy as t
    toy.m.category <--> t
    get.m.mover <--> toy.m
```

Example 2: Merging I-Get-Cup-Cn, You-Get-Cn

The learner can generalize to a person-Get-Cn. On the form side both constructions has an ordering constraint between the person and *get*. Additionally, both I-Cn and You-Cn are subcase of Ref-Expr. On the meaning side, Addressee and Speaker are both subcase of Human.

```
construction Person-Get-Cn
  level 2
  constructional
  constituents
    person : Ref-Expr
    get : Get-Cn
  form
    get.f before toy.f
  meaning
    evokes Human as h
    person.m.category <--> h
    get.m.causer <--> person.m
```